Question: I am trying to convert **uint** to **string** in solidity. Is it possible to convert? if so how can I convert it?

Convertion to base 10 is prefered, but in hexadecimal might also be helpful.

Answer: 
Another option is to use a solution from Oraclize <https://github.com/oraclize/ethereum-api/blob/master/oraclizeAPI_0.5.sol>, it suits best for me:

**0.5 Compiler Version:**

```
    function uint2str(uint _i) internal pure returns (string memory _uintAsString) {
    if (_i == 0) {
        return "0";
    }
    uint j = _i;
    uint len;
    while (j != 0) {
        len++;
        j /= 10;
    }
    bytes memory bstr = new bytes(len);
    uint k = len - 1;
    while (_i != 0) {
        bstr[k--] = byte(uint8(48 + _i % 10));
        _i /= 10;
    }
    return string(bstr);
}

```

**Pre 0.5 Compiler Version:**

```
function uint2str(uint i) internal pure returns (string){
    if (i == 0) return "0";
    uint j = i;
    uint length;
    while (j != 0){
        length++;
        j /= 10;
    }
    bytes memory bstr = new bytes(length);
    uint k = length - 1;
    while (i != 0){
        bstr[k--] = byte(48 + i % 10);
        i /= 10;
    }
    return string(bstr);
}

```
You can convert the uint to bytes32 by using `bytes32 data = bytes32(u)` (uint is same uint256 [How to convert a uint256 type integer into a bytes32?](https://ethereum.stackexchange.com/questions/6498/how-to-convert-a-uint256-type-integer-into-a-bytes32))

Then use [How to convert a bytes32 to string](https://ethereum.stackexchange.com/questions/2519/how-to-convert-a-bytes32-to-string):

```
function bytes32ToString (bytes32 data) returns (string) {
    bytes memory bytesString = new bytes(32);
    for (uint j=0; j<32; j++) {
        byte char = byte(bytes32(uint(data) * 2 ** (8 * j)));
        if (char != 0) {
            bytesString[j] = char;
        }
    }
    return string(bytesString);
}

```